

## EBNF Grammar for C--

```
1. TranslationUnit = { [ "extern" ] TypeSpecifier, identifier,
                      ( [ "[", integer, "]" ],
                        { ",", identifier, [ "[", integer, "]" ] }, ";",
                        | "(", [ Parameter, { ",", Parameter } ], ")",
                        ( CompoundStatement | ";" )
                      )
                    }
;
2. TypeSpecifier   = "void"
                    | "int"
                    | "float"
;
3. Parameter       = TypeSpecifier, identifier, [ "[", "]" ]
;
4. CompoundStatement = "{", { Declaration | Statement }, "}"
;
5. Declaration     = TypeSpecifier, identifier, [ "[", integer, "]" ],
                    { ",", identifier, [ "[", integer, "]" ] }, ";"
;
6. Statement       = ExpressionStatement
                    | CompoundStatement
                    | SelectionStatement
                    | RepetitionStatement
                    | ReturnStatement
;
7. ExpressionStatement = [ Expression, [ "=", Expression ] ], ";"
;
8. SelectionStatement = "if", "(", Expression, ")", Statement, [ "else",
                        Statement ]
;
9. RepetitionStatement = "while", "(", Expression, ")", Statement
;
10. ReturnStatement  = "return", [ Expression ], ";"
;
11. Expression       = AndExpression,
                    { "||", AndExpression }
;
12. AndExpression    = RelationExpression,
                    { "&&", RelationExpression }
;
13. RelationExpression = SimpleExpression,
                    [ ( "<=" | "<" | ">=" | ">" | "==" | "!=" ),
                      SimpleExpression ]
;
14. SimpleExpression = Term, { ( "+" | "-" ), Term }
;
15. Term             = Factor, { ( "*" | "/" | "%" ), Factor }
;
16. Factor           = { "+" | "-" | "!" }, Value
;
17. Value            = "(", Expression, ")"
                    | identifier,
                      [ "[", Expression, "]"
                      | "(", [ Expression, { ",", Expression } ], ")",
                      ]
                    | integer
                    | float
                    | string
;
;
```